



## Editorial

# Solving the Power Flow Problem in Transmission Networks Using Nonlinear Complex-Domain Modeling Via Julia Software

Oscar Danilo Montoya<sup>1</sup>, Edwin Rivas-Trujillo<sup>1</sup>, and Walter Gil-González<sup>2</sup>

<sup>1</sup>Universidad Distrital Francisco José de Caldas, Colombia 

<sup>2</sup>Universidad Tecnológica de Pereira 

Solving the power flow problem for transmission grids is crucial for ensuring the reliable and efficient operation of electrical power systems. Power flow analysis allows engineers to determine the voltage, current, and power flow of a network, which is essential for maintaining system stability and avoiding overloads. Accurate power flow solutions help to identify potential issues such as voltage drops, line losses, and system inefficiency, enabling the proactive maintenance and optimization of the network. This analysis is vital for integrating renewable energy sources, as it ensures effective power distribution even under variable generation conditions. Ultimately, solving the power flow problem enhances the overall resilience, reliability, and economic performance of transmission networks, supporting a stable supply of electricity to consumers.

## 1. Complex-variable power flow modeling

For electrical systems, the power flow problem can be formulated as a complex-variable optimization model (1). This optimization problem takes the following form:

**Objective function:**

$$\min P_{\text{loss}} = \text{Re} \left\{ \sum_{k \in \mathbf{N}} \sum_{m \in \mathbf{N}} \mathbb{V}_k^* \mathbb{Y}_{km} \mathbb{V}_m \right\}, \quad (1)$$

where  $P_{\text{loss}}$  represents the active power losses in the transmission system;  $\mathbb{V}_k$  and  $\mathbb{V}_m$  correspond to the voltage variables in the complex domain that are assignable to nodes  $k$  and  $m$ ;  $\mathbb{Y}_{km}$  is the component of the nodal admittance matrix that relates nodes  $k$  and  $m$ ; and  $\mathbf{N}$  represents the set that contains all the nodes in the system. Note that the operator  $\text{Re}\{\mathbb{X}\}$  obtains the real part of the complex variable  $\mathbb{X}$ , while the operator  $\mathbb{X}^*$  obtains its complex conjugate.

Editorial

© The authors;  
reproduction  
right holder  
Universidad  
Distrital  
Francisco José de  
Caldas.

Open access



**Set of constraints:**

$$\mathbb{S}_{g,k}^* - \mathbb{S}_{d,k}^* = \mathbb{V}_k^* \sum_{m \in \mathbf{N}} \mathbb{Y}_{km} \mathbb{V}_m, \{ \forall k \in \mathbf{N} \} \quad (2)$$

$$\mathbb{V}_s = V_s \angle 0, \quad s = \text{slack node} \quad (3)$$

$$|\mathbb{V}_j| = V_j, \quad j = \text{PV nodes} \quad (4)$$

$$\text{Re} \{ \mathbb{S}_{g,j} \} = P_{g,j}, \quad j = \text{PV nodes} \quad (5)$$

where  $\mathbb{S}_{g,k}$  corresponds to the complex power injected by the generator connected to node  $k$ ;  $\mathbb{S}_{d,k}$  represents the complex power demanded at node  $k$ ;  $\mathbb{V}_s$  denotes the complex voltage assigned to the slack node (reference bus), whose magnitude is  $V_s$ ;  $V_j$  corresponds to the magnitude of the voltage assignable to the nodes; and  $P_{g,j}$  is the active power assigned to these nodes.

It is important to mention that the formulation (1)–(5) can also represent the optimal power flow problem in the complex domain if the active power constraint assigned to the PV nodes is relaxed, allowing them to move within their maximum and minimum capacity limits (2).

## 2. The IEEE-WSCC

To illustrate the solution to the power flow problem in transmission systems according to the formulation (1)–(5), consider the IEEE-WSCC (Western System Coordinating Council) system, whose impedance and nodal voltage data are shown in Fig. 1. Note that the data reported in this figure were obtained by applying the Newton-Raphson method (3).




Figure 1. IEEE-WSCC (adapted from (4))

### 3. Julia software implementation

This section illustrates the computational implementation of the mathematical model (1)-(5) in the Julia software, using the Ipopt solver in the JuMP optimization environment (5).

```

using DataFrames, LinearAlgebra
Vb = 1; Sb = 100; Zb = 1;
branch_data = DataFrames.DataFrame([
(1,4,0.0000,0.0576,0.0000), (2,7,0.0000,0.0625,0.0000),
(3,9,0.0000,0.0586,0.0000), (4,5,0.0100,0.0850,0.1760),
(4,6,0.0170,0.0920,0.1580), (5,7,0.0320,0.1610,0.3060),
(6,9,0.0390,0.1700,0.3580), (7,8,0.0085,0.0720,0.1490),
(8,9,0.0119,0.1008,0.2090),]);
DataFrames.rename!(branch_data, [:k, :m, :Rkm, :Xkm, :Bk])
node_data = DataFrames.DataFrame([
(1, 3, 1.040, 0.0, 0.0, 0.0, 0.0),
(2, 2, 1.025, 0.0, 0.0, 163.0, 0.0),
(3, 2, 1.025, 0.0, 0.0, 85.0, 0.0),
(4, 0, 1.000, 0.0, 0.0, 0.0, 0.0),
(5, 0, 1.000, 125.0, 50.0, 0.0, 0.0),
(6, 0, 1.000, 90.0, 30.0, 0.0, 0.0),
(7, 0, 1.000, 0.0, 0.0, 0.0, 0.0),
(8, 0, 1.000, 100.0, 35.0, 0.0, 0.0),
(9, 0, 1.000, 0.0, 0.0, 0.0, 0.0),]);
DataFrames.rename!(node_data, [:k, :type, :Vk0,
:Pdk, :Qdk, :Pgk, :Qgk])
N = size(node_data,1); L = size(branch_data,1)
A = zeros(N,L)
for l = 1:L
    k = branch_data.k[l]; m = branch_data.m[l]
    A[k,l] = 1; A[m,l] = -1
end
z = (branch_data.Rkm .+ im*branch_data.Xkm)/Zb
Sd = (node_data.Pdk .+ im*node_data.Qdk)/Sb
Sgo = (node_data.Pgk .+ im*node_data.Qgk)/Sb
Ybus = A*inv(diagm(z))*transpose(A)
for l = 1:L
    k = branch_data.k[l]; m = branch_data.m[l]
    Ybus[k,k] = Ybus[k,k] + 1*im*branch_data.Bk[l]/2;
    Ybus[m,m] = Ybus[m,m] + 1*im*branch_data.Bk[l]/2;
end
using JuMP, Ipopt
PF = Model(Ipopt.Optimizer);
@variable(PF, Sg[k in 1:N] in ComplexPlane());
@variable(PF, V[k in 1:N] in ComplexPlane(),

```

```

        start = 1.0 + 0.0im);
for k in 1:N
    if node_data.type[k] == 3
        @constraint(PF,V[k] == node_data.Vk0[k] + 0*im);
    elseif node_data.type[k] == 2
        @constraint(PF,abs2(V[k])==(node_data.Vk0[k])^2);
        @constraint(PF,real(Sg[k]) == real(Sgo[k]));
    else
        @constraint(PF, Sg[k] == 0);
    end
    @constraint(PF,conj(Sg[k]) - conj(Sd[k]) ==
    conj(V[k])*sum(Ybus[k,m]*V[m] for m = 1:N))
end
@objective(PF,Min,Sb*real(sum(conj(V[k])*
sum(Ybus[k,m]*V[m] for m = 1:N) for k = 1:N)));
JuMP.optimize!(PF)
@show objective_value(PF);
bus_data = DataFrames.DataFrame(;k = 1:N,
Vmag = round.(abs.(value.(V)), digits = 4),
Vang = round.(angle.(value.(V))*180/pi, digits = 4),
Pg = round. (real(value.(Sg)),digits = 4),
Qg = round. (imag(value.(Sg)),digits = 4))

```

By executing this computational routine, the following results are obtained:

```

EXIT: Optimal Solution Found.
objective_value(PF) = 4.6410214744826453

```

Row	k	Vmag	Vang	Pg	Qg
1	1	1.04	0.0	0.7164	0.2705
2	2	1.025	9.28	1.63	0.0665
3	3	1.025	4.6648	0.85	-0.1086
4	4	1.0258	-2.2168	0.0	0.0
5	5	0.9956	-3.9888	-0.0	0.0
6	6	1.0127	-3.6874	-0.0	0.0
7	7	1.0258	3.7197	-0.0	-0.0
8	8	1.0159	0.7275	0.0	0.0
9	9	1.0324	1.9667	-0.0	0.0

Thus, it is possible to note that:

- i. The electrical variables, *i.e.*, the magnitude and angle of the voltages at each of the buses, are equivalent to the solution obtained using the Newton-Raphson method (Fig. 1) (3).
- ii. Under the given operating conditions, the active power losses for this system amount to 4.6410 MW.

**Remark 1.** Suppose that the programmer is interested in solving the optimal power flow problem for transmission networks. In this scenario, the active power injection at the PV buses is left unconstrained, allowing the programmer to determine the optimal combination of these power inputs in order to minimize the total grid power losses.

### Conclusion

This editorial note provides an easily implementable computational routine to deal with the power flow problem in transmission networks, which can be extended to any single-phase AC grid, using a complex-domain variable formulation. The solution to the power flow problem is obtained via the interior point optimizer (*i.e.*, the Ipopt solver) available in Julia's JuMP optimization environment. This routine is presented as a tutorial, using the IEEE-WSCC network to demonstrate that the power flow solution is equivalent to the solution reached with the classical Newton-Raphson method.

### Acknowledgment

This research received support from the Ibero-American Program of Science and Technology for Development (CYTED), through the thematic network 723RT0150, *i.e.*, *Red para la integración a gran escala de energías renovables en sistemas eléctricos* (RIBIERSE-CYTED).

## Referencias

- [1] O. D. Montoya, C. A. Ramírez-Vanegas, and J. R. González-Granada, "Dynamic active and reactive power compensation in distribution networks using pv-statcoms: A tutorial using the julia software," *Res. Eng.*, vol. 21, p. 101876, Mar. 2024. <https://doi.org/10.1016/j.rineng.2024.101876>↑ 1
- [2] X. Bai, H. Wei, K. Fujisawa, and Y. Wang, "Semidefinite programming for optimal power flow problems," *Int. J. Elec. Power Energy Sys.*, vol. 30, no. 6–7, pp. 383–392, Jul. 2008. <https://doi.org/10.1016/j.ijepes.2007.12.003>↑ 2
- [3] A. Garcés-Ruiz, W. J. Gil-González, and O. D. Montoya-Giraldo, *Introducción a la estabilidad de sistemas eléctricos de potencia*. Pereira, Colombia: Universidad Tecnológica de Pereira, 2023. <https://doi.org/10.22517/9789587228960>↑ 2, 4
- [4] P. Anderson and A. Fouad, *Power system control and stability*. Piscataway, NJ: Wiley-Interscience, 2003. <https://ieeexplore.ieee.org/servlet/opac?bknumber=5264012>↑ 2
- [5] J. Bezanson, A. Edelman, S. Karpinski, and V. B. Shah, "Julia: A fresh approach to numerical computing," *SIAM Rev.*, vol. 59, no. 1, pp. 65–98, 2017. <https://doi.org/10.1137/141000671>↑ 3

## Oscar Danilo Montoya

Compatibility and Electromagnetic Interference group, Department of Engineering, Universidad Distrital Francisco José de Caldas; Electrical Engineer, Master of Electrical Engineering, and PhD in Engineering.

**Email:** [odmontoyag@udistrital.edu.co](mailto:odmontoyag@udistrital.edu.co)

## Edwin Rivas-Trujillo

Compatibility and Electromagnetic Interference group, Department of Engineering, Universidad Distrital Francisco José de Caldas; Electrical Engineer, Master of Electrical Engineering, and Master and PhD in Electrical Engineering, Electronics, and Automation.

**Email:** [erivas@udistrital.edu.co](mailto:erivas@udistrital.edu.co)

## Walter Gil-González

Electromagnetic Fields and Energy Phenomena group, Department of Engineering, Universidad Tecnológica de Pereira; Electrical Engineer, Master of Electrical Engineering, PhD in Engineering, and PhD in Renewable Energy.

**Email:** [wjgil@utp.edu.co](mailto:wjgil@utp.edu.co)





**Available in:**

<https://www.redalyc.org/articulo.oa?id=498881644001>

How to cite

Complete issue

More information about this article

Journal's webpage in redalyc.org

Scientific Information System Redalyc  
Diamond Open Access scientific journal network  
Non-commercial open infrastructure owned by academia

Oscar Danilo Montoya, Edwin Rivas-Trujillo,  
Walter Gil-González

**Solving the Power Flow Problem in Transmission  
Networks Using Nonlinear Complex-Domain Modeling Via  
Julia Software**

*Ingeniería*

vol. 29, no. 3, e22697, 2024

Universidad Distrital Francisco José de Caldas,

**ISSN:** 0121-750X

**ISSN-E:** 2344-8393

**DOI:** <https://doi.org/10.14483/23448393.22697>